


**Example 5a: Mediation of Within-Person Fluctuation in Univariate MLM in STATA MIXED and R LMER Compared to Multivariate MLM in Mplus via Multilevel and Single-Level Structural Equation Modeling**  
(complete syntax, data, and output available for STATA, R, and Mplus electronically)

The limitations of univariate multilevel models (MLMs) can be addressed by switching to multivariate MLMs (via SEM or multilevel SEM), as in Mplus. The **primary difference is that rather than obtaining between and within effects through observed variable predictors, in multivariate MLMs the between and within parts of any level-1 predictor can be partitioned into level-2 random intercepts and level-1 residuals in the model**, the same as for the outcome in univariate MLMs. This example features multivariate MLMs in which a level-1 variable can be both a predictor and an outcome simultaneously, as is necessary in order to do multilevel mediational analysis of direct and indirect fixed effects. These models use the data from Hoffman (2015) chapter 8 examining relations of fluctuation across 5 days for 105 older adults in daily stressors, daily negative mood, and daily physical symptoms.

**Level-2, Between-Person (BP) Model:**


The result from multiplying the  $X \rightarrow M$  and  $M \rightarrow Y$  fixed effects together is called the **indirect effect**. This effect is the formal test of whether the  $X \rightarrow Y$  path differs before vs. after including  $M \rightarrow Y$ .

Because there are two levels of  $X \rightarrow M$  and  $M \rightarrow Y$  fixed effects, there are two levels of indirect effects—and mediation—too.



**Careful!** The level-2 fixed effects will be total BP effects *if the level-1 effect is specified directly as a fixed effect at level 1 only*, but the level-2 effects will be the contextual effects otherwise (as demonstrated here).

**Level-1, Within-Person (WP) Model:**



So long as slopes are included for their level-2 parts in the model (via observed person means or latent random intercepts), the level-1 fixed effects will be the WP effects (i.e., they will be unsmushed as needed).

**We will examine two options for how to include variables in these models:** (1) They can be treated as **observed predictors**, which is the same as in univariate MLM. This means that although the model estimates their fixed effects to predict the outcome(s), the predictors' means, variances, and covariances are *not* model parameters, and predictors do not have distributional assumptions. This also means that because they are *not* part of the model likelihood, **any rows (occasions) with missing predictors will be deleted**. (2) They can be treated as (latent) **outcomes**, either by predicting them with other variables, or just by letting the model estimate their variances and covariances at each applicable level (and mean at the highest level). So because **outcomes are part of the model likelihood, they can have missing data given their distributional assumptions**, such that any case that has at least one outcome will still be included. Using ML in Mplus, it is not possible to turn categorical predictors into outcomes when using the multilevel SEM syntax (although it does appear to be allowed using Bayes estimation instead within version 8.8, as shown in the electronic materials). For this reason, in the multivariate MLMs we will include our “X” daily stressor (0=no, 1=yes) as an observed level-1 predictor and its person mean (centered such that 0=.40) as an observed level-2 predictor. In contrast, our “M” daily negative mood and our “Y” daily physical symptoms will be outcomes whose variance is model-partitioned into latent variables (as depicted above).

**There are two ways of specifying level-1 fixed slopes in Mplus M-SEM, and they create different level-2 fixed slopes:** (1) If a level-1 fixed slope is specified directly in the level-1 %WITHIN% model, any level-2 fixed slope of the same variable will carry their total BP effects. (2) If the level-1 placeholder

syntax is used instead, such that the variable's level-1 fixed and level-2 random slope show up in the level-2 %BETWEEN% model—regardless of whether the random slope variance is estimated—then the variable's level-2 fixed slopes will instead carry their contextual effects. We will show both versions to illustrate this result, although based on previous analyses for these data, the **WP effects in this example will be fixed only**, as no random WP effects were significant. Further, we will also examine how to specify interactions in this multivariate MLM framework, which become **latent variable interactions** for which ML estimation requires numeric integration. Finally, there is no REML within Mplus, so **we will use ML for all models**. We will first examine the effects of X and M in predicting Y separately. Then, within a full mediation model, we will examine the  $X \rightarrow M$  effect and the unique effects of X and M in predicting Y.

### Step 1a: Fitting the Between-Person and Within-Person Stress (X) $\rightarrow$ Symptoms (Y) Effects (i.e., before controlling for M Negative Mood)

Univariate MLMs partitioning stress into level-1 WP vs. level-2 contextual effects by observed variables:

#### In STATA MIXED:

```
display "Step 1: X Stressors Predicting Symptoms Y"
mixed symptoms c.women c.age80 c.stressor c.PMstress40 ///
      c.women#c.age80, || PersonID: , mle nolog
lincom c.stressor*1 + c.PMstress40*1 // BP X to Y Effect
```

#### In R LMER:

```
print("Step 1: X Stressors Predicting Symptoms Y")
Step1 = lmer(data=Example4b, REML=FALSE,
             formula=symptoms~1+women+age80+
               stressor+PMstress40+women:age80+(1|PersonID))
print("Show results using Satterthwaite DDF")
summary(Step1, ddf="Satterthwaite")
print("BP X to Y Effect");
contest1D(Step1, ddf="Satterthwaite", L=c(0,0,0,1,1,0))
```

Note in Mplus: stressor = stress; PMstress40 = PMstr40

In Mplus, estimating the same Univariate MLM using M-SEM:

```
TITLE: Step 1: Predicting symptoms outcome from OBSERVED stress (so X --> Y)
DATA: FILE = Example5a.csv; ! No path needed if input and data in same folder
VARIABLE:
! List of ALL variables in stacked data file, in order
! Mplus does NOT know what they used to be called, though
NAMES = PersonID women age80 session symptoms mood2 PMmood2 stress PMstr40;
! List of ALL variables used in model (DEFINED variables at end)
USEVARIABLES = symptoms women age80 stress PMstr40 agesex;
! Missing data codes (here, -999)
MISSING = ALL (-999);
! Identify level-2 ID
CLUSTER = PersonID;
! Predictor variables with variation ONLY at level 1
WITHIN = stress;
! Predictor variables with variation ONLY at level 2
BETWEEN = age80 women agesex PMstr40;

DEFINE:      agesex = age80*women; ! Create observed level-2 interaction

ANALYSIS:    TYPE = TWOLEVEL RANDOM; ! 2-level model with random slopes
              ESTIMATOR = ML;         ! Can also use MLR for non-normality

MODEL:       ! X Stress --> Y Symptoms Model
              ! Level-1, Within-Person (WP) Model
%WITHIN%
  symptoms; ! L1 R: residual variance in symptoms
  WPXtoY | symptoms ON stress; ! Placeholder for L1 WP stress->symptoms

! Level-2, Person-Level Model;
%BETWEEN%
[symptoms]; ! Fixed intercept for symptoms
symptoms; ! L2 random intercept variance for symptoms
[WPXtoY] (WPXtoY); ! L1 WP fixed effect (label) of stress->symptoms
WPXtoY@0; ! L2 G: No random stress slope variance->symptoms

symptoms ON women (SextoY); ! BP fixed effect of women->symptoms
symptoms ON age80 (AgetoY); ! BP fixed effect of age->symptoms
symptoms ON agesex (AgesexY); ! BP fixed effect of age*women->symptoms
symptoms ON PMstr40 (conXtoY); ! Contextual fixed effect of stress->symptoms

MODEL CONSTRAINT: ! Linear combinations of fixed effects
  NEW(BPXtoY); ! Need to name each new created fixed effect
  BPXtoY = WPXtoY + conXtoY; ! BP effect of stress->symptoms
```

**Univariate MLM Results:** This is the exact same model in STATA MIXED, R LMER, and Mplus M-SEM (given ML estimation for all three programs, although p-values differ when using Satterthwaite DDF in R LMER) because both daily stressors and person mean stressors are treated as observed predictors, whereas symptoms is an outcome whose variance is partitioned into model-estimated latent variables.

AIC	BIC	logLik	deviance	df.resid
1424.4	1458.3	-704.2	1408.4	501

**Random effects:**

Groups	Name	Variance	Std.Dev.
PersonID	(Intercept)	0.83721	0.91499
Residual		0.61340	0.78320

Number of obs: 509, groups: PersonID, 105

**Fixed effects:**

	Estimate	Std. Error	df	t value	Pr(> t )
(Intercept)	1.586495	0.193743	115.159896	8.1887	0.00000000000004043
women	-0.518685	0.219907	105.308416	-2.3587	0.020186
age80	0.096764	0.033291	108.203337	2.9066	0.004432
stressor	0.110013	0.094868	403.459632	1.1596	0.246882
PMstress40	1.335160	0.301870	127.465958	4.4230	0.0000206452444176
women:age80	-0.106495	0.037894	107.128853	-2.8103	0.005886

```
> print("BP X to Y Effect")
```

```
[1] "BP X to Y Effect"
```

```
> contest1D(Step1, ddf = "Satterthwaite", L = c(0, 0, 0, 1, 1, 0))
```

	Estimate	Std. Error	df	t value	Pr(> t )
1	1.4451731	0.28643326	103.60072	5.0454096	0.0000019391296

**Univariate MLM Results in Mplus M-SEM:**

MODEL FIT INFORMATION

Number of Free Parameters	8
---------------------------	---

Loglikelihood	
H0 Value	-704.220

Information Criteria	
Akaike (AIC)	1424.440
Bayesian (BIC)	1458.299
Sample-Size Adjusted BIC	1432.906
(n* = (n + 2) / 24)	

MODEL RESULTS

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
Within Level				
Residual Variances				
SYMPTOMS	0.613	0.043	14.191	0.000
Between Level				
SYMPTOMS ON				
WOMEN	-0.519	0.220	-2.358	0.018
AGE80	0.097	0.033	2.906	0.004
AGESEX	-0.106	0.038	-2.810	0.005
PMSTR40	1.335	0.302	4.423	0.000
Means				
WPXTOY	0.110	0.095	1.159	0.246
Intercepts				
SYMPTOMS	1.586	0.194	8.188	0.000
Variances				
WPXTOY	0.000	0.000	999.000	999.000
Residual Variances				
SYMPTOMS	0.837	0.134	6.233	0.000
New/Additional Parameters				
BPSTRESS	1.445	0.286	5.046	0.000

## Step 2a: Fitting the Between-Person and Within-Person Mood (M) → Symptoms (Y) Effects (i.e., before controlling for X Symptoms)

Univariate MLMs partitioning mood into level-1 WP vs. level-2 contextual effects by observed variables:

### In STATA MIXED:

```
display "Step 2a-ish: Mood M Predicting Symptoms Y"
mixed symptoms c.women c.age80 c.mood2 c.PMmood2 ///
    c.women#c.age80, || PersonID: , mle nolog
lincom c.mood2*1 + c.PMmood2*1 // BP M to Y Effect
```

### In R LMER:

```
print("Step 2a-ish: M Mood Predicting Symptoms Y")
Step2 = lmer(data=Example4b, REML=FALSE,
    formula=symptoms~1+women+age80+
    mood2+PMmood2+women:age80+(1|PersonID))
print("Show results using Satterthwaite DDF")
summary(Step2, ddf="Satterthwaite")
print("BP M to Y Effect")
contestID(Step2, ddf="Satterthwaite", L=c(0,0,1,1,0))
```

**MLM Results:** Although this is the same idea, this is NOT the same model as in Mplus M-SEM (right), in which mood is treated like another outcome (and so its mean and level-specific variances are model parameters, even though it is not being predicted).

AIC	BIC	logLik	deviance	df.resid
1421.7	1455.5	-702.8	1405.7	501

Random effects:

Groups	Name	Variance	Std.Dev.
PersonID	(Intercept)	0.81615	0.90341
Residual		0.61273	0.78277

Number of obs: 509, groups: PersonID, 105

Fixed effects:

	Estimate	Std. Error	df	t value	Pr(> t )
(Intercept)	3.265483	0.345801	105.879511	9.4432	1.015e-15
women	-0.518135	0.217506	105.334759	-2.3822	0.01900
age80	0.066899	0.033494	107.758448	1.9973	0.04831
mood2	0.159103	0.127716	404.161789	1.2458	0.21358
PMmood2	1.810999	0.390993	132.127349	4.6318	8.572e-06
women:age80	-0.091762	0.037637	107.045874	-2.4381	0.01641

[1] "BP M to Y Effect"

	Estimate	Std. Error	df	t value	Pr(> t )
1	1.9701027	0.36873211	104.877	5.3429107	0.0000053576549

In multivariate Mplus, partitioning mood into WP vs. Contextual in the MODEL using placeholder syntax for level-1 effects (Model 2a):

**TITLE:** Step 2a: Predicting symptoms outcome from mood OUTCOME (so M --> Y)  
( DATA is the same )

### VARIABLE:

```
! List of ALL variables in stacked data file, in order
! Mplus does NOT know what they used to be called, though
NAMES = PersonID women age80 session symptoms mood2 PMmood2 stress PMstr40;
! List of ALL variables used in model (DEFINED variables at end)
USEVARIABLES = symptoms women age80 mood2 agesex;
! Missing data codes (here, -999)
MISSING = ALL (-999);
! Identify level-2 ID
CLUSTER = PersonID;
! Predictor variables with variation ONLY at level 1 -- none now
WITHIN = ;
! Predictor variables with variation ONLY at level 2 -- no PMmood2
BETWEEN = age80 women agesex;
```

( DEFINE and ANALYSIS are the same )

### MODEL: ! M Mood --> Y Symptoms Model

! Level-1, Within-Person (WP) Model

%WITHIN%

```
symptoms; ! L1 R: residual variance in symptoms
mood2; ! L1 R: residual variance in mood
WPMtoY | symptoms ON mood2; ! Placeholder for L1 WP mood->symptoms
```

! Level-2, Person-Level Model;

%BETWEEN%

```
[symptoms]; ! Fixed intercept for symptoms
symptoms; ! L2 random intercept variance in symptoms
[mood2]; ! Fixed intercept for mood
mood2; ! L2 random intercept variance in mood
```

```
[WPMtoY] (WPMtoY); ! L1 WP fixed effect of mood->symptoms
WPMtoY@0; ! L2 G: No rand mood slope var->symptoms
symptoms ON women (SextoY); ! BP fixed effect of women->symptoms
symptoms ON age80 (AgetoY); ! BP fixed effect of age->symptoms
symptoms ON agesex (AgesexY); ! BP fixed effect of age*women->symptoms
symptoms ON mood2 (conMtoY); ! Contextual fixed effect mood->symptoms
```

MODEL CONSTRAINT:

```
NEW(BPMtoY); ! Linear combinations of fixed effects
BPMtoY = WPMtoY + conMtoY; ! Name each new created fixed effect
! BP fixed effect of mood->symptoms
```

\*\*\* WARNING in MODEL command

In the MODEL command, the random regression predictor variable on the WITHIN level refers to the whole observed variable. To use the latent within-level part, use ESTIMATOR=BAYES in the ANALYSIS command.

**Mplus Multivariate Results using Placeholder Syntax:**  
underlined values indicate the 3 parameters for mood now  
 as an “outcome” not estimated in univariate MLM version

```
MODEL FIT INFORMATION
Number of Free Parameters          11

Loglikelihood
H0 Value                          -890.792

Information Criteria
Akaike (AIC)                      1803.583
Bayesian (BIC)                    1850.140
Sample-Size Adjusted BIC          1815.225
(n* = (n + 2) / 24)
```

**Model fit is the same either way**, but without placeholder syntax,  
 absolute fit tests also now appear, which are relative to a saturated  
 (unstructured) matrix of variances per level.

Let's see how the results differ based on the syntax:  
 bolded terms that are missing are noted in ( )

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
Within Level				
(SYMPTOMS ON MOOD2)				
Variances				
MOOD2	<u>0.093</u>	0.007	14.156	0.000
Residual Variances				
SYMPTOMS	0.613	0.043	14.185	0.000
Between Level				
SYMPTOMS ON				
WOMEN	-0.540	0.220	-2.458	0.014
AGE80	0.074	0.034	2.181	0.029
AGESEX	-0.098	0.038	-2.582	0.010
MOOD2	<b>2.340</b>	<b>0.558</b>	<b>4.196</b>	<b>0.000</b>
Means				
MOOD2	<u>-0.795</u>	0.026	-30.456	0.000
WPMTOY	<b>0.167</b>	<b>0.128</b>	<b>1.303</b>	<b>0.193</b>
Intercepts				
SYMPTOMS	3.710	0.463	8.020	0.000
Variances				
MOOD2	<u>0.052</u>	0.010	5.174	0.000
WPMTOY	<b>0.000</b>	<b>0.000</b>	<b>999.000</b>	<b>999.000</b>
Residual Variances				
SYMPTOMS	0.754	0.140	5.405	0.000
New/Additional Parameters				
BPMTOY	<b>2.506</b>	0.530	4.728	0.000

**Same model specifying level-1 fixed effect in %WITHIN% instead (Step 2b):**

( all previous commands are the same )

MODEL: ! M Mood --> Y Symptoms Model WITHOUT THE LEVEL-1 PLACEHOLDER

! Level-1, Within-Person (WP) Model

%WITHIN%

symptoms;

! L1 R: residual variance in symptoms

mood2;

! L1 R: residual variance in mood

symptoms ON mood2 (WPMtoY);

! NO Placeholder, L1 WP mood->symptoms here

! Level-2, Person-Level Model

%BETWEEN%

[symptoms];

! Fixed intercept for symptoms

symptoms;

! L2 random intercept variance in symptoms

[mood2];

! Fixed intercept for mood

Mood2;

! L2 random intercept variance in mood

! References to fixed and random effects of L1 WP mood are gone

symptoms ON women (SextoY);

! BP fixed effect of women->symptoms

symptoms ON age80 (AgetoY);

! BP fixed effect of age->symptoms

symptoms ON agesex (AgesexY);

! BP fixed effect of age\*women->symptoms

symptoms ON mood2 (BPMtoY);

! NOW BP fixed effect of mood->symptoms

MODEL CONSTRAINT:

! Linear combinations of fixed effects

NEW(conMtoY);

! Name each new created fixed effect

conMtoY = BPMtoY - WPMtoY;

! Contextual fixed effect of mood->symptoms

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
Within Level				
SYMPTOMS ON				
MOOD2	<b>0.167</b>	<b>0.128</b>	<b>1.303</b>	<b>0.193</b>
Variances				
MOOD2	<u>0.093</u>	0.007	14.157	0.000
Residual Variances				
SYMPTOMS	0.613	0.043	14.185	0.000
Between Level				
SYMPTOMS ON				
WOMEN	-0.540	0.220	-2.458	0.014
AGE80	0.074	0.034	2.181	0.029
AGESEX	-0.098	0.038	-2.582	0.010
MOOD2	<b>2.506</b>	0.530	4.727	0.000
Means				
MOOD2	<u>-0.795</u>	0.026	-30.454	0.000
(WPMTOY)				
Intercepts				
SYMPTOMS	3.710	0.463	8.020	0.000
Variances				
MOOD2	<u>0.052</u>	0.010	5.174	0.000
(WPMTOY)				
Residual Variances				
SYMPTOMS	0.754	0.140	5.405	0.000
New/Additional Parameters				
CONMTOY	<b>2.339</b>	0.558	4.195	0.000

**Step 3a: Fitting the Full Mediation Model: Between-Person and Within-Person Stress (X) → Mood (M) → Symptoms (Y)**

For parallel interpretation of the level-2 fixed effects of stress, the sex, age, and their interaction predictors also now predict mood.

A full simultaneous mediation model is not possible in univariate MLM, so here is Multivariate Mplus using placeholder syntax

→ WP + Contextual effects:

```

TITLE: Step3: Full mediation MLM of Stress --> Mood --> Symptoms
( DATA is the same )
VARIABLE:
! List of ALL variables in stacked data file, in order
  NAMES = PersonID women age80 session symptoms mood2 PMmood2
         stress PMstr40;
! List of ALL variables used in model (DEFINED variables at end)
  USEVARIABLES = symptoms women age80 mood2 stress PMstr40 agesex;
! Missing data codes (here, -999)
  MISSING = ALL (-999);
! Identify level-2 ID
  CLUSTER = PersonID;
! Predictor variables with variation ONLY at level 1
  WITHIN = stress;
! Predictor variables with variation ONLY at level 2
  BETWEEN = age80 women agesex PMstr40;
( DEFINE and ANALYSIS are the same )

MODEL:      ! Full X Stress --> M Mood --> Y Symptoms Mediation Model
! Level-1, Within-Person (WP) Model
%WITHIN%
  symptoms;          ! L1 R: residual variance in symptoms
  mood2;             ! L1 R: residual variance in mood
  WPXtoM | mood2     ON stress; ! Placeholder L1 WP stress->mood
  WPXtoY | symptoms ON stress; ! Placeholder L1 WP stress->symptoms
  WPMtoY | symptoms ON mood2;  ! Placeholder L1 WP mood->symptoms

! Level-2, Person-Level Model
%BETWEEN%
[symptoms];          ! Fixed intercept for symptoms
  symptoms;          ! L2 random intercept variance in symptoms
[mood2];             ! Fixed intercept for mood
  mood2;             ! L2 random intercept variance in mood
[WPXtoM] (WPXtoM);   ! L1 WP fixed effect of stress->mood
  WPXtoM@0;          ! L2 G: No random stress slope variance->mood
[WPXtoY] (WPXtoY);   ! L1 WP fixed effect of stress->symptoms
  WPXtoY@0;          ! L2 G: No random stress slope variance->symptoms
[WPMtoY] (WPMtoY);   ! L1 WP fixed effect of mood->symptoms
  WPMtoY@0;          ! L2 G: No random mood slope variance->symptoms

symptoms mood2 ON women; ! BP fixed effects women->mood, symptoms
symptoms mood2 ON age80; ! BP fixed effects age->mood, symptoms
symptoms mood2 ON agesex; ! BP fixed effects age*women

mood2      ON PMstr40(conXtoM); ! Contextual fixed effect stress->mood
symptoms ON PMstr40(conXtoY); ! Contextual fixed effect stress->symptoms
symptoms ON mood2 (conMtoY); ! Contextual fixed effect mood->symptoms

```

```

! Getting BP fixed effects and all indirect effects
MODEL CONSTRAINT:
NEW(BPXtoM BPXtoY BPMtoY WPind Conind BPind);
! BP effects;
  BPXtoM = WPXtoM + conXtoM; ! BP effect stress->mood
  BPXtoY = WPXtoY + conXtoY; ! BP effect stress->symptoms
  BPMtoY = WPMtoY + conMtoY; ! BP effect of mood->symptoms
! Indirect effects;
  WPind = WPXtoM*WPMtoY;      ! WP indirect effect
  Conind = conXtoM*conMtoY;    ! Contextual indirect effect
  BPind = BPXtoM*BPMtoY;      ! BP indirect effect

```

**Note:** MODEL INDIRECT is the usual way of obtaining indirect effects in Mplus but is not available for multilevel models. So we are using MODEL CONSTRAINT to calculate the indirect effects ourselves to accomplish the same thing. Further, although one can get bootstrapped *p*-values and confidence intervals for single-level mediation models, they are not available for multilevel mediation models. That means the *p*-values from the indirect effects may be a little suspect, and other methods of assessing significance may be needed for “best practice” (see [Kris Preacher’s website for online tools for bootstrapping parameter estimates](#)).

**Mplus Multivariate Results:**

## MODEL FIT INFORMATION

Number of Free Parameters	18
Loglikelihood	
H0 Value	-864.198
Information Criteria	
Akaike (AIC)	1764.396
Bayesian (BIC)	1840.580
Sample-Size Adjusted BIC	1783.446
(n* = (n + 2) / 24)	

MODEL RESULTS				
	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
Within Level				
Residual Variances				
SYMPTOMS	0.612	0.043	14.184	0.000
MOOD2	0.089	0.006	14.146	0.000
Between Level				
SYMPTOMS ON				
WOMEN	-0.534	0.209	-2.553	0.011
AGE80	0.070	0.033	2.121	0.034
AGESEX	-0.094	0.036	-2.596	0.009
PMSTR40	1.091	0.304	3.589	0.000
MOOD2	1.852	0.606	3.058	0.002
MOOD2 ON				
WOMEN	0.008	0.054	0.151	0.880
AGE80	0.013	0.008	1.629	0.103
AGESEX	-0.006	0.009	-0.628	0.530
PMSTR40	0.124	0.079	1.561	0.119
Means				
WPXTOM	0.162	0.036	4.486	0.000
WPXTOY	0.085	0.097	0.872	0.383
WPMTOY	0.141	0.131	1.077	0.281
Intercepts				
SYMPTOMS	3.340	0.540	6.184	0.000
MOOD2	-0.880	0.049	-17.879	0.000
Variances				
WPXTOM	0.000	0.000	999.000	999.000
WPXTOY	0.000	0.000	999.000	999.000
WPMTOY	0.000	0.000	999.000	999.000
Residual Variances				
SYMPTOMS	0.678	0.122	5.547	0.000
MOOD2	0.040	0.008	4.802	0.000
New/Additional Parameters				
BPXTOM	0.286	0.070	4.063	0.000
BPXTOY	1.175	0.289	4.067	0.000
BPMTOY	1.993	0.576	3.459	0.001
WPIND	0.023	0.022	1.048	0.295
CONIND	0.229	0.164	1.393	0.164
BPIND	0.570	0.217	2.630	0.009

### Step 4a: Same Model, Adding Mood\*Sex Interactions → Symptoms

When I tried to estimate a latent variable interaction between level-2 observed variable women and level-2 random intercept mood2, Mplus insisted that was an observed variable interaction, which would instead be between original level-1 mood and women. So I had to create a work-around that renamed the mood random intercept:

( all previous commands are the same after adding ALGORITHM = INTEGRATION)

```
MODEL:      ! X Stress --> M Mood --> Y Symptoms Mediation Model + Mood*Sex
! Level-1, Within-Person (WP) Model
%WITHIN%
  symptoms;      ! L1 R: residual variance in symptoms
  mood2;         ! L1 R: residual variance in mood
  WPXtoM | mood2 ON stress; ! Placeholder L1 WP stress->mood
  WPXtoY | symptoms ON stress; ! Placeholder L1 WP stress->symptoms
  WPMtoY | symptoms ON mood2; ! Placeholder L1 WP mood->symptoms

! Level-2, Person-Level Model;
%BETWEEN%
  [symptoms];      ! Fixed intercept for symptoms
  symptoms;        ! L2 random intercept variance in symptoms

  moodint BY mood2@1; ! Rename mood random intercept as latent variable
[moodint mood2@0]; ! Fixed intercept for moodint, not mood
  moodint mood2@0; ! L2 G: random intercept variance for moodint, not mood
! Now moodint replaces mood2 everywhere in the syntax below

[WPXtoM] (WPXtoM); ! L1 WP fixed effect of stress->mood
  WPXtoM@0;      ! L2 G: No random stress slope variance->mood
[WPXtoY] (WPXtoY); ! L1 WP fixed effect of stress->symptoms
  WPXtoY@0;      ! L2 G: No random stress slope variance->symptoms
[WPMtoY] (WPMtoY); ! L1 WP fixed effect of mood->symptoms
  WPMtoY@0;      ! L2 G: No random mood slope variance->symptoms

symptoms moodint ON women; ! BP fixed effects women->mood, symptoms
symptoms moodint ON age80; ! BP fixed effects age->mood, symptoms
symptoms moodint ON agesex; ! BP fixed effects age*women

moodint ON PMstr40(conXtoM); ! Contextual fixed effect stress->mood
symptoms ON PMstr40(conXtoY); ! Contextual fixed effect stress->symptoms
symptoms ON moodint (conMtoY); ! Contextual effect of mood->symptoms

WPMtoY ON women (WPMsexY); ! Level-1 mood by sex->symptoms
moodsex | women XWITH moodint; ! Latent interaction of sex*context mood
symptoms ON moodsex (conMsexY); ! Contextual mood*sex->symptoms

MODEL CONSTRAINT:
( all previous new effects stayed here )
NEW(BPMsexY);
  BPMsexY = WPMsexY + conMsexY; ! BP mood*sex->symptoms
```

### Step 4a: Same Model, Adding Mood\*Sex Interactions → Symptoms

When I tried to estimate a latent variable interaction between level-2 observed variable women and level-2 random intercept mood2, Mplus insisted that was an observed variable interaction, which would instead be between original level-1 mood and women. So I had to create a work-around that renamed the mood random intercept:

( all previous commands are the same after adding ALGORITHM = INTEGRATION)

```

MODEL:      ! X Stress --> M Mood --> Y Symptoms Mediation Model + Mood*Sex
! Level-1, Within-Person (WP) Model
%WITHIN%
  symptoms;          ! L1 R: residual variance in symptoms
  mood2;             ! L1 R: residual variance in mood
  WPXtoM | mood2      ON stress; ! Placeholder L1 WP stress->mood
  WPXtoY | symptoms ON stress; ! Placeholder L1 WP stress->symptoms
  WPMtoY | symptoms ON mood2;  ! Placeholder L1 WP mood->symptoms

! Level-2, Person-Level Model;
%BETWEEN%
  [symptoms];        ! Fixed intercept for symptoms
  symptoms;          ! L2 random intercept variance in symptoms

moodint BY mood2@1; ! Rename mood random intercept as latent variable
[moodint mood2@0];  ! Fixed intercept for moodint, not mood
moodint mood2@0;    ! L2 G: random intercept variance for moodint, not mood
! Now moodint replaces mood2 everywhere in the syntax below

[WPXtoM] (WPXtoM); ! L1 WP fixed effect of stress->mood
WPXtoM@0;          ! L2 G: No random stress slope variance->mood
[WPXtoY] (WPXtoY); ! L1 WP fixed effect of stress->symptoms
WPXtoY@0;          ! L2 G: No random stress slope variance->symptoms
[WPMtoY] (WPMtoY); ! L1 WP fixed effect of mood->symptoms
WPMtoY@0;          ! L2 G: No random mood slope variance->symptoms

symptoms moodint ON women; ! BP fixed effects women->mood, symptoms
symptoms moodint ON age80; ! BP fixed effects age->mood, symptoms
symptoms moodint ON agesex; ! BP fixed effects age*women

moodint ON PMstr40(conXtoM); ! Contextual fixed effect stress->mood
symptoms ON PMstr40(conXtoY); ! Contextual fixed effect stress->symptoms
symptoms ON moodint (conMtoY); ! Contextual effect of mood->symptoms

WPMtoY ON women (WPMsexY); ! Level-1 mood by sex->symptoms
moodsex | women XWITH moodint; ! Latent interaction of sex*context mood
symptoms ON moodsex (conMsexY); ! Contextual mood*sex->symptoms

MODEL CONSTRAINT:
( all previous new effects stayed here )
NEW(BPMsexY);
  BPMsexY = WPMsexY + conMsexY; ! BP mood*sex->symptoms

```

**Step 4a Multivariate Mplus Results (a few minutes later):****New effects are in bold**

Number of Free Parameters				20
Loglikelihood				
H0 Value				-862.992
Information Criteria				
Akaike (AIC)				1765.984
Bayesian (BIC)				1850.633
Sample-Size Adjusted BIC				1787.150
(n* = (n + 2) / 24)				
MODEL RESULTS				
	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
Within Level				
Residual Variances				
SYMPTOMS	0.611	0.043	14.191	0.000
MOOD2	0.090	0.006	14.095	0.000
Between Level				
MOODINT BY				
MOOD2	1.000	0.000	999.000	999.000
MOODINT ON				
WOMEN	0.006	0.054	0.119	0.905
AGE80	0.014	0.008	1.706	0.088
AGESEX	-0.006	0.009	-0.689	0.491
PMSTR40	0.140	0.079	1.787	0.074
<b>WPMTOY ON</b>				
<b>WOMEN</b>	<b>0.107</b>	<b>0.198</b>	<b>0.542</b>	<b>0.588</b>
SYMPTOMS ON				
MOODINT	4.016	1.501	2.675	0.007
<b>MOODSEX</b>	<b>-2.394</b>	<b>1.531</b>	<b>-1.564</b>	<b>0.118</b>
SYMPTOMS ON				
WOMEN	-2.529	1.325	-1.909	0.056
AGE80	0.040	0.041	0.965	0.335
AGESEX	-0.063	0.044	-1.422	0.155
PMSTR40	0.987	0.310	3.180	0.001
Means				
WPXTOM	0.156	0.036	4.309	0.000
WPXTOY	0.085	0.097	0.881	0.378
Intercepts				
SYMPTOMS	5.151	1.299	3.964	0.000
MOOD2	0.000	0.000	999.000	999.000
MOODINT	-0.876	0.049	-17.888	0.000
WPMTOY	0.053	0.201	0.261	0.794

Variances				
WPXTOM	0.000	0.000	999.000	999.000
WPXTOY	0.000	0.000	999.000	999.000
Residual Variances				
SYMPTOMS	0.625	0.123	5.088	0.000
MOOD2	0.000	0.000	999.000	999.000
MOODINT	0.039	0.008	4.738	0.000
WPMTOY	0.000	0.000	999.000	999.000
New/Additional Parameters				
BPXTOM	0.296	0.070	4.237	0.000
BPXTOY	1.072	0.295	3.628	0.000
BPMTOY	4.068	1.478	2.753	0.006
WPIND	0.008	0.031	0.260	0.795
CONIND	0.564	0.394	1.433	0.152
BPIND	1.205	0.535	2.253	0.024
<b>BPMSEXY</b>	<b>-2.287</b>	<b>1.509</b>	<b>-1.516</b>	<b>0.130</b>



### Example Results Section for Steps 1–3:

The relationships among time-varying stressors (i.e., whether or not a stressor was reported on a given day), negative mood (constructed as the mean of five items), and physical symptoms (constructed as the sum of five reported symptoms) were examined using multivariate multilevel models (i.e., multilevel structural equation modeling) within Mplus v. 8.11 (Muthén & Muthén, 1998-2017) using maximum likelihood (ML) estimation. (We obtained an identical pattern of results using a robust ML estimator to account for potential non-normality, so the original ML results are reported below.) Two observed variables were used to partition the effect of binary daily stressors (0=no, 1=yes) into its contextual (level-2; incremental between-person) and within-person (level-1) effects, in which the level-2 predictor was created as the person mean of stressors centered at 40% of days ( $PMstress_i - .40$ ) and the level-1 predictor was daily stressors. An empty model ICC in the logit metric was .448, indicating about half of the variance in daily stressors was between persons. Variance partitioning was accomplished within the model estimation instead for the more continuous level-1 outcomes of negative mood ( $ICC = .360$ ) and physical symptoms ( $ICC = .659$ ), such that random intercept variances were estimated for each at level 2, and residual variances were estimated for each at level 1. Under this specification, level-1 fixed effects indicate within-person effects, whereas level-2 fixed effects reflect contextual effects. The MODEL CONSTRAINT command was used to obtain model-implied between-person effects and all indirect effects. Age, sex, and their interaction (with 80-year-old men as the reference group) were included as predictors in the level-2 model for both negative mood and physical symptoms. In addition, likelihood ratio revealed no significant random within-person direct effects in any of the models (all  $-2\Delta LL(\sim 2) < 5.99, p > .05$ ), so all within-person direct effects were fixed across persons. Although our eventual goal was to examine the extent to which negative mood mediated the between-person and within-person effects of stressors on physical symptoms, we began by estimating separate models for stress and mood each predicting symptoms before controlling for each effect for the other.

First, a univariate multilevel model of observed stressors predicting physical symptoms ( $X \rightarrow Y$ ) revealed significant positive contextual (1.335) and between-person (1.445) effects but no significant within-person effect. These first results indicate that, after controlling for age and sex but before controlling for negative mood, physical symptoms were higher on average for persons who experienced more stressor days than others (even after controlling for daily stressors for the contextual effect), but physical symptoms on a given day were not related to whether a stressor was experienced that day. Second, a separate multivariate multilevel model of negative mood predicting physical symptoms ( $M \rightarrow Y$ ) revealed significant contextual (2.339) and between-person (2.506) effects but no significant within-person effect. These second results indicate that, after controlling for age and sex but before controlling for stressors, physical symptoms were higher on average for persons who reported higher negative mood than others (even after controlling for daily negative mood for the contextual effect), but physical symptoms on a given day were not related to whether a negative mood was higher than usual that day. Thus, to summarize, significant direct effects were found between persons (at level 2) for both  $X \rightarrow Y$  and  $M \rightarrow Y$ , but no significant direct effects were found within persons.

Third, the extent to which daily negative mood mediated the relationship between daily stressors and daily physical symptoms at each level was examined in a multilevel mediation model with all three variables, each specified as previously described. For comparable interpretation of the level-2 effects of stressors on mood and symptoms, level-2 effects of age, sex, and their interaction were added to predict negative mood (as well as symptoms, as before). Results are shown in Table X. At level 2, although there was a significant positive between-person effect (0.286) of observed stressors predicting negative mood ( $X \rightarrow M$ ), the corresponding contextual effect (0.124) was not significant, indicating that negative mood was not significantly higher in persons with more stressor days after controlling for the presence of a daily stressor. In addition, the between-person effect of stressors on physical symptoms ( $X \rightarrow Y$ ) was significantly reduced (from 1.445 to 1.175) after controlling for the between-person effect of negative mood, as indicated by a significant between-person indirect effect of stressors on physical symptoms through negative mood. Likewise, the between-person effect of negative mood on physical symptoms ( $M \rightarrow Y$ ) was reduced (from 2.506 to 1.852) after controlling for daily stressors. Both between-person effects of stressors and negative mood predicting symptoms (and their contextual effects) remained uniquely significant. Thus, reporting more stressor days than others was related to reporting more physical symptoms than others (even after controlling for daily stressors), but this link did not result solely from a concomitant difference in negative mood. However, the contextual indirect effect was not significant, indicating that some of this mediation is reduced after controlling for daily stressors and daily negative mood. At level 1, there was a significant  $X \rightarrow M$  within-person effect (0.162), indicating that greater stressors than usual on a given day did predict greater negative mood than usual that day. However,

the within-person effect of stressors on physical symptoms ( $X \rightarrow Y$ ) was not significantly reduced (and was still not significant) after controlling for negative mood, as indicated by a nonsignificant within-person indirect effect of stressors on physical symptoms through negative mood. Thus, after controlling for people's general tendencies to report stressors, reporting a stressor that day did not predict reporting more physical symptoms that day. Finally, the within-person effect of negative mood on symptoms ( $M \rightarrow Y$ ) remained nonsignificant after controlling for daily stressors as well.

Equation for Step 3 (using placeholder syntax in ML, the full level-1 outcome is used as a predictor, not just the within-person latent level-1 residual):

$$\text{Level 1: } Mood_{ti} = \beta_{0iM} + \beta_{1iM}(Stress_{ti}) + e_{tiM}$$

$$\text{Level 1: } Symptoms_{ti} = \beta_{0iY} + \beta_{1iY}(Stress_{ti}) + \beta_{2iY}(Mood_{ti}) + e_{tiY}$$

Level 2 Mood:

$$\beta_{0iM} = \gamma_{00M} + \gamma_{01M}(Age_i - 80) + \gamma_{02M}(Women_i) + \gamma_{04M}(Women_i)(Age_i - 80) + \gamma_{05M}(PMstressor_i - .40) + U_{0iM}$$

$$\beta_{1iM} = \gamma_{10M}$$

Level 2 Symptoms:

$$\beta_{0iY} = \gamma_{00Y} + \gamma_{01Y}(Age_i - 80) + \gamma_{02Y}(Women_i) + \gamma_{04Y}(Women_i)(Age_i - 80) + \gamma_{05Y}(PMstressor_i - .40) + \gamma_{06Y}(\beta_{0iM}) + U_{0iY}$$

$$\beta_{1iY} = \gamma_{10Y}$$

$$\beta_{2iY} = \gamma_{20Y}$$

**Bonus Step 5: Fitting the Full Mediation Model via SEM: Between-Person and Within-Person Stress (X) → Mood (M) → Symptoms (Y)**  
**Level-1 stress now must be treated as an outcome, which means this model is not equivalent to the previous Step 3a in MLM**

```

TITLE: Step 5: SEM Full Mediation Model using Stress Intercept Factor
DATA: FILE = Example5a.csv; ! No path if input and data in same folder

! Unstacking to multivariate format
DATA LONGTOWIDE:
! Names of old stacked former variables (without numbers)
LONG = stress|mood|symptom;
! Names of new multivariate variables (that use numbers)
WIDE = stress1-stress5|mood1-mood5|symptom1-symptom5;
! Variable with level-2 ID info
IDVARIABLE = PersonID;
! Old level-1 identifier
REPETITION = session (2 3 4 5 6);

VARIABLE:
! List of ALL variables in stacked data file, in order
! Mplus does NOT know what they used to be called, though
NAMES = PersonID women age80 session symptom mood PMmood2
      stress PMstr40;
! List of ALL variables used in model (DEFINED variables at end)
USEVARIABLES = women age80 stress1-stress5 mood1-mood5
      symptom1-symptom5 agesex;
! Missing data codes (here, -999)
MISSING = ALL (-999);
! Identify stress as binary outcome
CATEGORICAL = stress1-stress5;

DEFINE: agesex = age80*women; ! Create observed level-2 interaction
ANALYSIS: ESTIMATOR = ML; MODEL = NOCOVARIANCES;
      INTEGRATION = MONTECARLO(1000);

MODEL: ! X = stress, M = mood, Y = symptoms
! All variable thresholds and intercepts fixed to 0
[stress1$1-stress5$1@0 mood1-mood5@0 symptom1-symptom5@0];
mood1-mood5 (Mresvar); ! L1 R: M residual variances held equal
symptom1-symptom5 (Yresvar); ! L1 R: Y residual variances held equal

! Define L2 intercept latent factors for each
FXint BY stress1-stress5@1;
FMint BY mood1-mood5@1;
FYint BY symptom1-symptom5@1;
! Fixed intercepts estimated
[FXint FMint FYint];
! L2 G: Random intercept variances estimated
FXint FMint FYint;

! L2 fixed effects of age and sex
FYint FMint ON women; ! BP fixed effects women->mood, symptoms
FYint FMint ON age80; ! BP fixed effects age->mood, symptoms
FYint FMint ON agesex; ! BP fixed effects age*women

! L2 mediation model
FMint ON FXint (conXtoM); ! Contextual effect stress->mood
FYint ON FXint (conXtoY); ! Contextual effect stress->symptoms
FYint ON FMint (conMtoY); ! Contextual effect mood->symptoms

! L1 WP fixed effect stress->mood
mood1-mood5 PON stress1-stress5 (WPXtoM);
! L1 WP fixed effect stress->symptoms
symptom1-symptom5 PON stress1-stress5 (WPXtoY);
! L1 WP fixed effect mood->symptoms
symptom1-symptom5 PON mood1-mood5 (WPMtoY);

! Getting BP total fixed effects and all indirect effects
MODEL CONSTRAINT:
NEW(BPXtoM BPXtoY BPMtoY WPind Conind BPind);
! BP effects;
BPXtoM = WPXtoM + conXtoM; ! BP effect stress->mood
BPXtoY = WPXtoY + conXtoY; ! BP effect stress->symptoms
BPMtoY = WPMtoY + conMtoY; ! BP effect of mood->symptoms
! Indirect effects;
WPind = WPXtoM*WPMtoY; ! WP indirect effect
Conind = conXtoM*conMtoY; ! Contextual indirect effect
BPind = BPXtoM*BPMtoY; ! BP indirect effect

```

**Note:** We are again using MODEL CONSTRAINT to calculate the indirect effects ourselves. Further, although one can get bootstrapped  $p$ -values and confidence intervals for single-level mediation models, they are not available for multilevel mediation models. That means the  $p$ -values from the indirect effects may be a little suspect, and other methods of assessing significance may be needed for “best practice” (see [Kris Preacher’s website for online tools for bootstrapping parameter estimates](#)).

### Mplus SEM Results:

Number of Free Parameters	20
Loglikelihood	
H0 Value	-1180.753
Information Criteria	
Akaike (AIC)	2401.505
Bayesian (BIC)	2454.585
Sample-Size Adjusted BIC	2391.401
(n* = (n + 2) / 24)	

**Step 5 Results (a few minutes later): Different effects are in bold  
2 new effects are underlined**

		Estimate	S.E.	Est./S.E.	Two-Tailed P-Value	
Factor loadings fixed to 1 are omitted						
FMINT	ON					
FXINT		0.038	0.021	1.798	0.072	
FYINT	ON					
<b>FXINT</b>		<b>0.264</b>	<b>0.088</b>	<b>3.005</b>	<b>0.003</b>	<b>context</b>
FMINT		1.597	0.637	2.506	0.012	
FYINT	ON					
WOMEN		-0.519	0.210	-2.476	0.013	
AGE80		0.072	0.033	2.201	0.028	
AGESEX		-0.096	0.036	-2.635	0.008	
FMINT	ON					
WOMEN		0.010	0.054	0.192	0.848	
AGE80		0.013	0.008	1.614	0.106	
AGESEX		-0.006	0.009	-0.620	0.536	
MOOD1	ON					
STRESS1		0.156	0.036	4.277	0.000	<b>X -&gt; M</b>
MOOD2	ON					
STRESS2		0.156	0.036	4.277	0.000	
MOOD3	ON					
STRESS3		0.156	0.036	4.277	0.000	
MOOD4	ON					
STRESS4		0.156	0.036	4.277	0.000	
MOOD5	ON					
STRESS5		0.156	0.036	4.277	0.000	
SYMPTOM1	ON					
STRESS1		0.093	0.097	0.955	0.340	<b>X -&gt; Y</b>
MOOD1		0.141	0.131	1.077	0.282	<b>M -&gt; Y</b>
SYMPTOM2	ON					
STRESS2		0.093	0.097	0.955	0.340	
MOOD2		0.141	0.131	1.077	0.282	
SYMPTOM3	ON					
STRESS3		0.093	0.097	0.955	0.340	
MOOD3		0.141	0.131	1.077	0.282	
SYMPTOM4	ON					
STRESS4		0.093	0.097	0.955	0.340	
MOOD4		0.141	0.131	1.077	0.282	
SYMPTOM5	ON					
STRESS5		0.093	0.097	0.955	0.340	
MOOD5		0.141	0.131	1.077	0.282	

Means				
FXINT	-0.256	0.194	-1.319	0.187
Intercepts				
Intercepts fixed to 0 are omitted				
FMINT	-0.863	0.050	-17.096	0.000
FYINT	3.221	0.558	5.769	0.000
Thresholds				
STRESS1\$1	0.000	0.000	999.000	999.000
STRESS2\$1	0.000	0.000	999.000	999.000
STRESS3\$1	0.000	0.000	999.000	999.000
STRESS4\$1	0.000	0.000	999.000	999.000
STRESS5\$1	0.000	0.000	999.000	999.000
Variances				
FXINT	2.599	0.720	3.608	0.000
Residual Variances				
MOOD1	0.089	0.006	14.151	0.000
MOOD2	0.089	0.006	14.151	0.000
MOOD3	0.089	0.006	14.151	0.000
MOOD4	0.089	0.006	14.151	0.000
MOOD5	0.089	0.006	14.151	0.000
SYMPTOM1	0.612	0.043	14.184	0.000
SYMPTOM2	0.612	0.043	14.184	0.000
SYMPTOM3	0.612	0.043	14.184	0.000
SYMPTOM4	0.612	0.043	14.184	0.000
SYMPTOM5	0.612	0.043	14.184	0.000
FMINT	0.038	0.008	4.559	0.000
FYINT	0.637	0.128	4.970	0.000
New/Additional Parameters				
BPXTOM	0.194	0.033	5.942	0.000
BPXTOY	0.356	0.110	3.231	0.001
BPMTOY	1.738	0.609	2.854	0.004
WPIND	0.022	0.021	1.044	0.297
CONIND	0.061	0.038	1.606	0.108
BPIND	0.337	0.129	2.615	0.009
<b>Previous results from MLM treating stress as observed:</b>				
Means				
WPXTOM	0.162	0.036	4.486	0.000
WPXTOY	0.085	0.097	0.872	0.383
WPMTOY	0.141	0.131	1.077	0.281
New/Additional Parameters:				
BPXTOM	0.286	0.070	4.063	0.000
BPXTOY	1.175	0.289	4.067	0.000
BPMTOY	1.993	0.576	3.459	0.001
WPIND	0.023	0.022	1.048	0.295
CONIND	0.229	0.164	1.393	0.164
BPIND	0.570	0.217	2.630	0.009